Q:

Select Two Cells in a While Cycle

Greetings to all!
I'm new at VBA, and i'm doing an application and i am a little bit confused in a certain part of a code, please find example in the table below:

My code needs to create a folder based on each row cells, in example:
Starting with B1, it validates if it is Father or Son, if it is Father, saves the first letter (in this case "F") and if it is "Son", it's saves an "S".
Second, if the A1, is 100, it must save the cell value.
And now the thing what i need to do is create a Folder in the ActiveWorkbook.path which names it (as in the last example) \F100
 \S101, \S102 and goes on..
My code is something like this but i know that it doesn't work corretly.
Sub Test()

Dim rng1 As Range
Dim rng2 As Range

Set rng1 = Range("A1")
Set rng2 = Range("B1").Select

    Set fso = CreateObject("Scripting.FileSystemObject")

Do While Not IsEmpty(rng2)

    If rng2.Value = "Father" Then
        fso.createFolder (ActiveWorkbook.Path & "\" & rng1.Value2)

        ElseIf rng2.Value = "Son" Then
        fso.createFolder (ActiveWorkbook.Path & "\" & rng1.Value2)

    End If

    ActiveCell.Offset(1, 0).Select

Loop

End Sub

Thanks for the time !

A:

Something like:
Sub Test()

    Dim rng1 As Range, fso, v

    Set rng1 = Range("A1")
    Set fso = CreateObject("Scripting.FileSystemObject")

    Do While Not IsEmpty(rng1)

        v = rng1.Offset(0, 1).value

        If v = "Father" Or v = "Son" Then

            fso.createFolder (ActiveWorkbook.Path & "\" & _
                              Left(v, 1) & rng1.Value2)

        End If

        Set rng1 = rng1.Offset(1, 0)

    Loop

End Sub

